Supplementary Information

Selective detection of ^{13}CH signals from the mixture of $^{13}CH/^{13}CH_2/^{13}CH_3$ groups in biomolecules with enhanced sensitivity

Tairan Yuwen, Zhilian Xia, Jiangshu Liu, Yixin Cui, Jitendra Das, Youlin Xia, Paolo Rossi, Charalampos G. Kalodimos

Department of Structural Biology, St. Jude Children's Research Hospital, Memphis, TN 38105, USA

Supplementary figures



Fig. S1 The intensities of different spin orders versus τ_b or τ_{mix} in (A) INEPT-based and (B) CP-based coherence transfer scheme for $^{13}CH_2$ group with the initial density matrix corresponding to $2C_zH_z$ and H_z, respectively. In Panel A the magnitudes of generated spin orders C_z and $4C_zH_zH_z$ are identical. $^1J_{CH}$ = 125 Hz was used in the numerical simulation, and relaxation effects were not included.


(A)
(B)
τb = 1.19 ms
τb = 1.18 ms
τb = 1.17 ms
τb = 1.16 ms
τb = 1.15 ms
τb = 1.14 ms
τb = 1.13 ms
τb = 1.12 ms
τb = 1.11 ms
τmix = 6.7 ms
τmix = 6.6 ms
τmix = 6.5 ms
τmix = 6.4 ms
τmix = 6.3 ms
τmix = 6.2 ms
τmix = 6.1 ms
τmix = 6.0 ms
τmix = 5.9 ms
2.0 1.5 1.0 0.5 0.0 −0.5
2.0 1.5 1.0 0.5 0.0 −0.5
1H [ppm]
1H [ppm]

Fig. S2 One-dimensional version of CH-selective $^{1}H–^{13}C$ HSQC (unenhanced implementation) measured for $^{13}CH_3$-labeled protein L with INEPT-based (Panel A) and CP-based schemes (Panel B). A series of τ_b or τ_{mix} around the optimal value (τ_b = 1.15 ms or τ_{mix} = 6.3 ms) were used, which was determined based on signal intensities of $^{13}CH_3$ groups. The residual $^{13}CH_3$ signals are negative when τ_b or τ_{mix} is shorter than the optimal value, whereas positive when τ_b or τ_{mix} is longer than the optimal value.


(A)
(B)
τb = 1.13 ms
τb = 1.12 ms
τb = 1.11 ms
τb = 1.10 ms
τb = 1.09 ms
τb = 1.08 ms
τb = 1.07 ms
τb = 1.06 ms
τb = 1.05 ms
τmix = 6.4 ms
τmix = 6.3 ms
τmix = 6.2 ms
τmix = 6.1 ms
τmix = 6.0 ms
τmix = 5.9 ms
τmix = 5.8 ms
τmix = 5.7 ms
τmix = 5.6 ms
2.0 1.5 1.0 0.5 0.0 −0.5
1H [ppm]
2.0 1.5 1.0 0.5 0.0 −0.5
1H [ppm]

Fig. S3 One-dimensional version of CH-selective $^{1}H–^{13}C$ HSQC (unenhanced implementation) measured for $^{13}CH_3$-labeled L99A T4L with INEPT-based (Panel A) and CP-based schemes (Panel B). A series of τ_b or τ_{mix} around the optimal value (τ_b = 1.09 ms or τ_{mix} = 6.0 ms) were used, which was determined based on signal intensities of $^{13}CH_3$ groups. The residual $^{13}CH_3$ signals are negative when τ_b or τ_{mix} is shorter than the optimal value, whereas positive when τ_b or τ_{mix} is longer than the optimal value.

Pulse sequence code (Bruker)

```
;Chsqc.G5.ty

;1. replace all pulse p2  with p58:sp58. spnam58='G5'
;2. replace all pulse p4  with p60:sp60. spnam60='G5'

;Chsqc
;for shorter pulse sequence and better signal
;modified from hsqcetgpsisp2.2 by youlin Xia
;plw9 may be 0 to turn off presaturation

;This sequence is modified to suppress methyl CH3 signals
;use -DCH_sel to selectively detect CH groups
;use -Dcp_new or -Dcp_old to achieve better suppression of methyl CH3 signals
;use -Dwater_flg to get better water suppression
;use -Dwater_flg or presaturation with -Dcp_new or -Dcp_old is recommended
;use DIGMOD = baseopt to get more flat spectral baselines
;longer d1 should be used with stronger het-cp power level
;-Dcp_new allows using any het-cp mixing time, but may not work with old version of To
pspin
;-Dcp_old only allows using integer number of DIPSI2 cycle

;setup experiment by chsqc or type following commands
;rpar HSQCETGPSISP.2 all
;pulprog chsqc
;1 FnMODE States
;o1p 4.696
;getprosol



;hsqcetgpsisp2.2
;avance-version (07/04/04)
;HSQC
;2D H-1/X correlation via double inept transfer
;   using sensitivity improvement
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;with decoupling during acquisition
;using trim pulses in inept transfer
;using shaped pulses for all 180degree pulses on f2 - channel
;with gradients in back-inept
;
;A.G. Palmer III, J. Cavanagh, P.E. Wright & M. Rance, J. Magn.
;   Reson. 93, 151-170 (1991)
;L.E. Kay, P. Keifer & T. Saarinen, J. Am. Chem. Soc. 114,
;   10663-5 (1992)
;J. Schleucher, M. Schwendinger, M. Sattler, P. Schmidt, O. Schedletzky,
;   S.J. Glaser, O.W. Sorensen & C. Griesinger, J. Biomol. NMR 4,
;   301-306 (1994)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple>


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>


"in0=inf1/2"

#ifdef f1180
  "d0=(in0/2)+3u"
#else
```

```
  "d0=3u"
#endif

"d11=30m"
"d12=20u"

"d4=1s/(abs(cnst2)*4.0)"
"d5=1s/(abs(cnst3)*4.0)"


"p30=5m"

"p58=p1*10.0"    /* Universal PI pulse  */
"spoal58=0.5"     /* Universal PI pulse offset */
"spoffs58=0.0"     /* Universal PI pulse offset */
"spw58=plw1"      /* Universal PI pulse power level  */
;spnam58: G5,   (p58:sp58 ph1)

"p60=p3*10.0"    /* Universal PI pulse  */
"spoal60=0.5"     /* Universal PI pulse offset */
"spoffs60=0.0"     /* Universal PI pulse offset */
"spw60=plw2"      /* Universal PI pulse power level  */
;spnam60: G5,   (p60:sp60 ph1):f2

"DELTA1=d4-4u-p19-d16"
"DELTA2=d6-4u-p19-d16"
"DELTA3=p58+6u"

#ifdef cp_old
  "d8=(2590.0/90.0)*p13*l2"
#endif

#if defined(cp_new) || defined(cp_old)
  "plw13=plw1*pow(p1/p13,2)"
  "plw23=plw2*pow(p3/p13,2)"
  "cnst4=1s/(p13*4.0)"
#endif

#ifdef water_flg
  "spoffs11=0"
  "spoal11=1"
#endif

#ifdef mess_flg
  define pulse dly_pg1     /* Messerle purge pulse */
    "dly_pg1=5m"

  define pulse dly_pg2     /* Messerle purge pulse */
    "dly_pg2=dly_pg1/1.62"
#endif

"acqt0=-p1*2/PI"


1 d11 ze
  d12 pl12:f2

#if defined(cp_new) || defined(cp_old)
  if "cnst4 > 20000 || cnst4 < 5000" {
    2u
    print "error: power level for het-cp too strong or too weak; check p13 !!!"
    goto HaltAcqu
  }

  if "d8 > d5*6.0" {
    2u
    print "error: het-cp transfer time too long; check d8 or l2 !!!"
    goto HaltAcqu
  }
```

```
#ifndef CH_sel
    2u
    print "error: -Dcp_new or -Dcp_old should be used together with -DCH_sel !!!"
    goto HaltAcqu
#endif
#endif

#if defined(cp_new) && defined(cp_old)
    2u
    print "error: -Dcp_new and -Dcp_old should not be used together !!!"
    goto HaltAcqu
#endif

#ifdef Ddec
; d11 LOCKDEC_ON         /* Not required for AvanceIII-HD */
  50u LOCKH_ON
  d11 H2_PULSE
#endif

2 d11 do:f2
  2u rpp11 rpp12
  d12 fq=0(sfo hz):f1
  d12 pl9:f1

#ifdef Ddec
  "DELTA = d1 - 300m - d11 - 6m"

  300m cw:f1 ph1
  4u do:f1
  d11 H2_LOCK
  6m LOCKH_OFF
  DELTA cw:f1 ph1
  4u do:f1
  d12 pl1:f1

  50u LOCKH_ON
  15u H2_PULSE
  50u UNBLKGRAMP
#else /*Ddec*/
  d1 cw:f1 ph1
  4u do:f1
  d12 pl1:f1
  50u UNBLKGRAD
#endif /*Ddec*/

  d12 pl2:f2

  (p3 ph1):f2
  4u
  p16:gp1*0.8
  d16
  (p3 ph2):f2
  4u
  p16:gp1
  d16

#ifdef water_flg
  (p11:sp11 ph1):f1

  4u
  p30:gp1*2.0
  d16
#endif

  d12 fq=cnst1(bf ppm):f1
  d12 pl1:f1

  (p1 ph1):f1
```

```
#ifdef CH_sel
#if defined(cp_new) || defined(cp_old)
  "DELTA=min(d12, d13)"
  DELTA pl13:f1 pl23:f2

#ifdef cp_new
  d8 cpds1:f1 cpds1:f2 ph2
  DELTA do:f1 do:f2
#else /*cp_new*/
4 (center (p13*3.556 ph11):f1 (p13*3.556 ph11):f2)
  (center (p13*4.556 ph12):f1 (p13*4.556 ph12):f2)
  (center (p13*3.222 ph11):f1 (p13*3.222 ph11):f2)
  (center (p13*3.167 ph12):f1 (p13*3.167 ph12):f2)
  (center (p13*0.333 ph11):f1 (p13*0.333 ph11):f2)
  (center (p13*2.722 ph12):f1 (p13*2.722 ph12):f2)
  (center (p13*4.167 ph11):f1 (p13*4.167 ph11):f2)
  (center (p13*2.944 ph12):f1 (p13*2.944 ph12 ipp12):f2)
  (center (p13*4.111 ph11):f1 (p13*4.111 ph11 ipp11):f2)
  lo to 4 times l2
#endif /*cp_new*/

  DELTA pl1:f1 pl2:f2
  (p3 ph5):f2
#else /*cp_flg*/
  4u
  p19:gp3
  d16
  DELTA1
  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )
  4u
  p19:gp3
  d16
  DELTA1 pl1:f1 pl2:f2
  (p1 ph0):f1

  4u
  p16:gp2
  d16

#ifdef Ddec
  d12 pl4:f4
  (p53 ph2):f4
  d12 pl17:f4
  (2u cpds4 ph1):f4
#endif

  (p3 ph1):f2

#ifdef Ddec
  d6
#else
  4u
  p19:gp3
  d16
  DELTA2
#endif

  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )

#ifdef Ddec
  d6 pl1:f1 pl2:f2
#else
  4u
  p19:gp3
  d16
  DELTA2 pl1:f1 pl2:f2
#endif

  (p3 ph2):f2
```

```
#ifdef Ddec
  4u do:f4
  d12 pl4:f4
  (p53 ph0):f4
#endif
#endif /*cp_flg*/

#ifdef mess_flg
  d12 fq=0(sfo hz):f1
  d12 pl11:f1
  (dly_pg1 ph1):f1
  4u
  (dly_pg2 ph2):f1
  d12 fq=cnst1(bf ppm):f1
  d12 pl1:f1
#endif
#else /*CH_sel*/
  4u
  p19:gp3
  DELTA1
  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )
  4u
  p19:gp3
  DELTA1 pl1:f1 pl2:f2
  (p1 ph0):f1
#endif /*CH_sel*/

  4u
  p16:gp4
  d16

#ifdef Ddec
  d12 pl4:f4
  (p53 ph2):f4
  d12 pl17:f4
  (2u cpds4 ph1):f4
#endif

  (p3 ph3):f2
#ifdef CH_sel
  if "d5 >= d0" {
    "DELTA = d5 - (d0 - 3u)"
    DELTA

    (center (p58:sp58 ph1):f1  (p60:sp60 ph4:r):f2 )

    "DELTA = d5 + (d0 - 3u)"
  }
  else {
    "DELTA = p58"
    DELTA

    (p60:sp60 ph4:r):f2

    "DELTA = (d0 - 3u) - (d5 - p58*0.5)"
    DELTA

    (p58:sp58 ph1):f1

    "DELTA = (d0 - 3u) + (d5 - p58*0.5)"
  }
  DELTA pl1:f1 pl2:f2

  (p3 ph2):f2
#else
  d0
  (p58:sp58 ph1):f1
  d0
```

```
  (p60:sp60 ph4:r):f2
  DELTA3 pl1:f1 pl2:f2
  (p3 ph1):f2
#endif

#ifdef Ddec
  4u do:f4
  d12 pl4:f4
  (p53 ph0):f4
#endif

  4u
  p16:gp5
  d16

  (p1 ph1):f1
  4u
  p19:gp3
  d16
  DELTA1
  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )
  4u
  p19:gp3
  d16
  DELTA1 pl1:f1
  (p1 ph2):f1

  4u
  p16:gp6
  d16

#ifdef Ddec
  4u BLKGRAMP
#else
  4u BLKGRAD
#endif

  d12 pl12:f2

  (p1 ph1):f1

  go=2 ph31 cpds2:f2
  d11 do:f2 mc #0 to 2
     F1PH(calph(ph3, +90), caldel(d0, +in0) & calph(ph3, +180) & calph(ph31, +180))

#ifdef Ddec
  d11 H2_LOCK
  d11 LOCKH_OFF
; d11 LOCKDEC_OFF        /* use statement for earlier hardware */
#endif

HaltAcqu, 1m
exit


ph0=3
ph1=0
ph2=1
ph3=0 2
ph4=0 0 1 1 2 2 3 3
ph5=2
ph11=1 3 3 1
ph12=3 1 1 3
ph31=0 2 2 0


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl4: dpwr3 - power level for 2H pulse pwd (p53)
```

```
;pl9 : f1 channel - power level for presaturation
;pl11: f1 channel - power level for Messerle purge, ~1 Watt
;pl12: f2 channel - power level for CPD/BB decoupling
;pl17: power level for 2H waltz decoupling during t1 and taub
;sp11: f1 channel - power level for water selective pulse pw_sl (away from z-axis)
;spnam11: shape for pw_sl (away from z-axis)
;p1 : f1 channel - 90 degree high power pulse
;p3 : f2 channel - 90 degree high power pulse
;p11: f1 channel - 90 degree shaped pulse for pw_sl (water selective)
;p13: f1/f2 channels - 90 degree pulse for het-cp, 15~25 us
;p16: homospoil/gradient pulse                          [1000 usec]
;p19: gradient pulse 2                                  [500 usec]
;p30: gradient pulse 3                                  [5 msec]
;p53: f4 channel - 90 degree high power pulse (pwd)
;d0 : incremented delay (F1 in 3D)                      [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : INEPT transfer time, 1.1~1.2 ms for J(CH) = 125 Hz, need to optimize
;d8 : het-cp transfer time, 5.0~6.0 ms for J(CH) = 125 Hz, need to optimize
;d11: delay for disk I/O                                [30 msec]
;d12: delay for power switching                         [20 usec]
;d13: very short delay for power switching              [0.2 usec]
;d16: delay for homospoil/gradient recovery             [200 usec]
;cnst1: center of aliphatic 1H chemical shifts (offset, in ppm)
;cnst2: 1H-13C J-coupling constant of CH groups (in Hz)
;cnst3: 1H-13C J-coupling constant of CH2 (or CH) groups (in Hz)
;cnst4: power level for 1H-13C het-cp (in Hz)
;l2: loop for DIPSI2 cycle: d8 = ((p13*28.778) * l2)
;inf1: 1/SW(C) = 2 * DW(C)
;in0: 1/(2 * SW(C)) = DW(C)
;nd0: 2
;ns: 2 * n
;ds: 32
;td1: number of experiments in F1
;FnMODE: States in F1
;cpdprg1: dipsi2.p13
;cpd2: decoupling according to sequence defined by cpdprg2
;cpdprg4: 2H decoupling sequence applied during t1 and taub [waltz16]
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;pcpd4: f4 channel - 90 degree pulse for 2H decoupling sequence


;use gradient ratio:    gp 1 : gp 2 : gp 3 : gp 4: gp 5: gp 6
;                         30 :   65 :   20 :  -60:   60:   40

;for z-only gradients:
;gpz1: 30%
;gpz2: 65%
;gpz3: 20%
;gpz4: -60%
;gpz5: 60%
;gpz6: 40%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100
;gpnam6: SMSQ10.100


;$Id: hcchdigp3d2,v 1.12 2014/02/11 11:31:28 ber Exp $
```

```
;Chsqc.G5.ty2

;1. replace all pulse p2  with p58:sp58. spnam58='G5'
;2. replace all pulse p4  with p60:sp60. spnam60='G5'

;Chsqc
;for shorter pulse sequence and better signal
;modified from hsqcetgpsisp2.2 by youlin Xia
;plw9 may be 0 to turn off presaturation

;This sequence is modified to suppress methyl CH3 signals
;use -DCH_sel to selectively detect CH groups
;use -Dcp_new or -Dcp_old to achieve better suppression of methyl CH3 signals
;use -Dwater_flg to get better water suppression
;use -Dwater_flg or presaturation with -Dcp_new or -Dcp_old is recommended
;longer d1 should be used with stronger het-cp power level
;-Dcp_new allows using any het-cp mixing time, but may not work with old version of To
pspin
;-Dcp_old only allows using integer number of DIPSI2 cycle

;setup experiment by chsqc or type following commands
;rpar HSQCETGPSISP.2 all
;pulprog chsqc
;1 FnMODE Echo-AntiEcho
;o1p 4.696
;getprosol



;hsqcetgpsisp2.2
;avance-version (07/04/04)
;HSQC
;2D H-1/X correlation via double inept transfer
;   using sensitivity improvement
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;with decoupling during acquisition
;using trim pulses in inept transfer
;using shaped pulses for all 180degree pulses on f2 - channel
;with gradients in back-inept
;
;A.G. Palmer III, J. Cavanagh, P.E. Wright & M. Rance, J. Magn.
;   Reson. 93, 151-170 (1991)
;L.E. Kay, P. Keifer & T. Saarinen, J. Am. Chem. Soc. 114,
;   10663-5 (1992)
;J. Schleucher, M. Schwendinger, M. Sattler, P. Schmidt, O. Schedletzky,
;   S.J. Glaser, O.W. Sorensen & C. Griesinger, J. Biomol. NMR 4,
;   301-306 (1994)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple>


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>


"in0=inf1/2"

#ifdef f1180
  "d0=(in0/2)+3u"
#else
  "d0=3u"
```

```
#endif

"d11=30m"
"d12=20u"

"d4=1s/(abs(cnst2)*4.0)"
"d5=1s/(abs(cnst3)*4.0)"


"p30=5m"

"p58=p1*10.0"    /* Universal PI pulse  */
"spoal58=0.5"     /* Universal PI pulse offset */
"spoffs58=0.0"     /* Universal PI pulse offset */
"spw58=plw1"      /* Universal PI pulse power level  */
;spnam58: G5,   (p58:sp58 ph1)

"p60=p3*10.0"    /* Universal PI pulse  */
"spoal60=0.5"     /* Universal PI pulse offset */
"spoffs60=0.0"     /* Universal PI pulse offset */
"spw60=plw2"      /* Universal PI pulse power level  */
;spnam60: G5,   (p60:sp60 ph1):f2

"DELTA1=d4-4u-p19-d16"
"DELTA2=d6-4u-p19-d16"

#ifdef cp_old
  "d8=(2590.0/90.0)*p13*l2"
#endif

#if defined(cp_new) || defined(cp_old)
  "plw13=plw1*pow(p1/p13,2)"
  "plw23=plw2*pow(p3/p13,2)"
  "cnst4=1s/(p13*4.0)"
#endif

#ifdef cp_back
  "plw14=plw1*pow(p1/p14,2)"
  "plw26=plw2*pow(p3/p14,2)"
#endif

#ifdef water_flg
  "spoffs11=0"
  "spoal11=1"
#endif

#ifdef mess_flg
  define pulse dly_pg1     /* Messerle purge pulse */
    "dly_pg1=5m"

  define pulse dly_pg2     /* Messerle purge pulse */
    "dly_pg2=dly_pg1/1.62"
#endif

"acqt0=0"
baseopt_echo


1 d11 ze
  d12 pl12:f2

#if defined(cp_new) || defined(cp_old)
  if "cnst4 > 20000 || cnst4 < 5000" {
    2u
    print "error: power level for het-cp too strong or too weak; check p13 !!!"
    goto HaltAcqu
  }

  if "d8 > d5*6.0" {
```

```
    2u
    print "error: het-cp transfer time too long; check d8 or l2 !!!"
    goto HaltAcqu
  }

#ifndef CH_sel
    2u
    print "error: -Dcp_new or -Dcp_old should be used together with -DCH_sel !!!"
    goto HaltAcqu
#endif
#endif

#if defined(cp_new) && defined(cp_old)
    2u
    print "error: -Dcp_new and -Dcp_old should not be used together !!!"
    goto HaltAcqu
#endif

#if defined(cp_back) && defined(CH_sel)
    2u
    print "warning: -Dcp_back and -DCH_sel should not be used together for suppressing
 13CH3 groups !!!"
#endif

#ifdef Ddec
; d11 LOCKDEC_ON          /* Not required for AvanceIII-HD */
  50u LOCKH_ON
  d11 H2_PULSE
#endif

2 d11 do:f2
  2u rpp11 rpp12
  d12 fq=0(sfo hz):f1
  d12 pl9:f1

#ifdef Ddec
  "DELTA = d1 - 300m - d11 - 6m"

  300m cw:f1 ph1
  4u do:f1
  d11 H2_LOCK
  6m LOCKH_OFF
  DELTA cw:f1 ph1
  4u do:f1
  d12 pl1:f1

  50u LOCKH_ON
  15u H2_PULSE
  50u UNBLKGRAMP
#else /*Ddec*/
  d1 cw:f1 ph1
  4u do:f1
  d12 pl1:f1
  50u UNBLKGRAD
#endif /*Ddec*/

  d12 pl2:f2

  (p3 ph1):f2
  4u
  p16:gp1*0.8
  d16
  (p3 ph2):f2
  4u
  p16:gp1
  d16

#ifdef water_flg
  (p11:sp11 ph1):f1
```

```
  4u
  p30:gp1*2.0
  d16
#endif

  d12 fq=cnst1(bf ppm):f1
  d12 pl1:f1

  (p1 ph1):f1
#ifdef CH_sel
#if defined(cp_new) || defined(cp_old)
  "DELTA=min(d12, d13)"
  DELTA pl13:f1 pl23:f2

#ifdef cp_new
  d8 cpds1:f1 cpds1:f2 ph2
  DELTA do:f1 do:f2
#else /*cp_new*/
4 (center (p13*3.556 ph11):f1 (p13*3.556 ph11):f2)
  (center (p13*4.556 ph12):f1 (p13*4.556 ph12):f2)
  (center (p13*3.222 ph11):f1 (p13*3.222 ph11):f2)
  (center (p13*3.167 ph12):f1 (p13*3.167 ph12):f2)
  (center (p13*0.333 ph11):f1 (p13*0.333 ph11):f2)
  (center (p13*2.722 ph12):f1 (p13*2.722 ph12):f2)
  (center (p13*4.167 ph11):f1 (p13*4.167 ph11):f2)
  (center (p13*2.944 ph12):f1 (p13*2.944 ph12 ipp12):f2)
  (center (p13*4.111 ph11):f1 (p13*4.111 ph11 ipp11):f2)
  lo to 4 times l2
#endif /*cp_new*/

  DELTA pl1:f1 pl2:f2
  (p3 ph6):f2
#else /*cp_flg*/
  4u
  p19:gp3
  d16
  DELTA1
  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )
  4u
  p19:gp3
  d16
  DELTA1 pl1:f1 pl2:f2
  (p1 ph0):f1

  4u
  p16:gp2
  d16

#ifdef Ddec
  d12 pl4:f4
  (p53 ph2):f4
  d12 pl17:f4
  (2u cpds4 ph1):f4
#endif

  (p3 ph1):f2

#ifdef Ddec
  d6
#else
  4u
  p19:gp3
  d16
  DELTA2
#endif

  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )
```

```
#ifdef Ddec
  d6 pl1:f1 pl2:f2
#else
  4u
  p19:gp3
  d16
  DELTA2 pl1:f1 pl2:f2
#endif

  (p3 ph2):f2

#ifdef Ddec
  4u do:f4
  d12 pl4:f4
  (p53 ph0):f4
#endif
#endif /*cp_flg*/

#ifdef mess_flg
  d12 fq=0(sfo hz):f1
  d12 pl11:f1
  (dly_pg1 ph1):f1
  4u
  (dly_pg2 ph2):f1
  d12 fq=cnst1(bf ppm):f1
  d12 pl1:f1
#endif
#else /*CH_sel*/
  4u
  p19:gp3
  DELTA1
  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )
  4u
  p19:gp3
  DELTA1 pl1:f1 pl2:f2
  (p1 ph0):f1
#endif /*CH_sel*/

  4u
#if defined(cp_new) || defined(cp_old)
  p16:gp4*-1
#else
  p16:gp4
#endif
  d16

  2u rpp11 rpp12

#ifdef Ddec
  d12 pl4:f4
  (p53 ph2):f4
  d12 pl17:f4
  (2u cpds4 ph1):f4
#endif

  (p3 ph3):f2

  if "p1 > p3" {
    "DELTA = p1-p3"
    DELTA
  }

#ifdef CH_sel
  if "d5 >= d0" {
    "DELTA = d5 - (d0 - 3u)"
    DELTA

    (center (p58:sp58 ph1):f1  (p60:sp60 ph4:r):f2 )
```

```
#ifdef Ddec
    "DELTA = d5 + (d0 - 3u) - (4u + d12 + p53) - (2u + p19 + d16)"
#else
    "DELTA = d5 + (d0 - 3u) - (2u + p19 + d16)"
#endif
  }
  else {
    "DELTA = p58"
    DELTA

    (p60:sp60 ph4:r):f2

    "DELTA = (d0 - 3u) - (d5 - p58*0.5)"
    DELTA

    (p58:sp58 ph1):f1

#ifdef Ddec
    "DELTA = (d0 - 3u) + (d5 - p58*0.5) - (4u + d12 + p53) - (2u + p19 + d16)"
#else
    "DELTA = (d0 - 3u) + (d5 - p58*0.5) - (2u + p19 + d16)"
#endif
  }
  DELTA pl1:f1 pl2:f2

#ifdef Ddec
  4u do:f4
  d12 pl4:f4
  (p53 ph0):f4
#endif

  2u
  p19:gp7*-1*EA
  d16
#else /*CH_sel*/
  d0
  (p58:sp58 ph1):f1
  d0

#ifdef Ddec
  4u do:f4
  d12 pl4:f4
  (p53 ph0):f4
#endif

  2u
  p19:gp7*EA*0.5
  d16

  (p60:sp60 ph4:r):f2

  2u
  p19:gp7*EA*-0.5
  d16

#ifdef Ddec
  "DELTA = p58 + 6u + 4u + d12 + p53"
#else
  "DELTA = p58 + 6u"
#endif
  DELTA pl1:f1 pl2:f2
#endif /*CH_sel*/

  (ralign (p1 ph1):f1 (p3 ph5):f2)
#ifdef cp_back
  "DELTA=min(d12, d13)"
  DELTA pl14:f1 pl26:f2
5 (center (p14*3.556 ph11):f1 (p14*3.556 ph11):f2)
  (center (p14*4.556 ph12):f1 (p14*4.556 ph12):f2)
```

```
  (center (p14*3.222 ph11):f1 (p14*3.222 ph11):f2)
  (center (p14*3.167 ph12):f1 (p14*3.167 ph12):f2)
  (center (p14*0.333 ph11):f1 (p14*0.333 ph11):f2)
  (center (p14*2.722 ph12):f1 (p14*2.722 ph12):f2)
  (center (p14*4.167 ph11):f1 (p14*4.167 ph11):f2)
  (center (p14*2.944 ph12):f1 (p14*2.944 ph12 ipp12):f2)
  (center (p14*4.111 ph11):f1 (p14*4.111 ph11 ipp11):f2)
  lo to 5 times 8
  DELTA pl1:f1
#else /*cp_back*/
  2u
  p19:gp5
  d16

#ifdef Ddec
  "DELTA = d4 - d12 - p53 - d12 - 2u - 2u - p19 - d16"

  d12 pl4:f4
  (p53 ph2):f4
  d12 pl17:f4
  (2u cpds4 ph1):f4
#else
  "DELTA = d4 - 2u - p19 - d16"
#endif
  DELTA

  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )

#ifdef Ddec
  "DELTA = d4 - 4u - d12 - p53 - 2u - p19 - d16"
  DELTA pl1:f1 pl2:f2

  4u do:f4
  d12 pl4:f4
  (p53 ph0):f4
#else
  "DELTA = d4 - 2u - p19 - d16"
  DELTA pl1:f1 pl2:f2
#endif

  2u
  p19:gp5
  d16

  (center (p1 ph2):f1 (p3 ph0):f2)

  2u
  p19:gp6
  d16
  DELTA1

  (center (p58:sp58 ph1):f1  (p60:sp60 ph1:r):f2 )

  2u
  p19:gp6
  d16

  if "p3 > p1" {
    "DELTA = (p3-p1)*0.5"
    DELTA
  }

  DELTA1 pl1:f1
#endif /*cp_back*/

  (p1 ph1):f1

  "DELTA = 2u + p19 + d16 + 4u + d12 + de - p1*2.0/PI"
  DELTA
```

```
  (p58:sp58 ph1):f1

  2u
#ifdef cp_back
  p19:gp8
#else
  p19:gp8*-1.0
#endif
  d16

#ifdef Ddec
  4u BLKGRAMP
#else
  4u BLKGRAD
#endif

  d12 pl12:f2

  go=2 ph31 cpds2:f2
  d11 do:f2 mc #0 to 2
     F1EA(calgrad(EA) & calph(ph5, +180), caldel(d0, +in0) & calph(ph3, +180) & calph(
ph31, +180))

#ifdef Ddec
  d11 H2_LOCK
  d11 LOCKH_OFF
; d11 LOCKDEC_OFF        /* use statement for earlier hardware */
#endif

HaltAcqu, 1m
exit


ph0=3
ph1=0
ph2=1
ph3=0 2
ph4=0 0 1 1 2 2 3 3
ph5=0
ph6=2
ph11=1 3 3 1
ph12=3 1 1 3
ph31=0 2 2 0


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl4: dpwr3 - power level for 2H pulse pwd (p53)
;pl9 : f1 channel - power level for presaturation
;pl11: f1 channel - power level for Messerle purge, ~1 Watt
;pl12: f2 channel - power level for CPD/BB decoupling
;pl17: power level for 2H waltz decoupling during t1 and taub
;sp11: f1 channel - power level for water selective pulse pw_sl (away from z-axis)
;spnam11: shape for pw_sl (away from z-axis)
;p1 : f1 channel - 90 degree high power pulse
;p3 : f2 channel - 90 degree high power pulse
;p11: f1 channel - 90 degree shaped pulse for pw_sl (water selective)
;p13: f1/f2 channels - 90 degree pulse for het-cp forward transfer, 15~25 us
;p14: f1/f2 channels - 90 degree pulse for het-cp back transfer, 34.5 us
;p16: homospoil/gradient pulse                         [1000 usec]
;p19: gradient pulse 2                                 [500 usec]
;p30: gradient pulse 3                                 [5 msec]
;p53: f4 channel - 90 degree high power pulse (pwd)
;d0 : incremented delay (F1 in 3D)                     [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : INEPT transfer time, 1.1~1.2 ms for J(CH) = 125 Hz, need to optimize
;d8 : het-cp transfer time, 5.0~6.0 ms for J(CH) = 125 Hz, need to optimize
;d11: delay for disk I/O                               [30 msec]
```

```
;d12: delay for power switching                        [20 usec]
;d13: very short delay for power switching             [0.2 usec]
;d16: delay for homospoil/gradient recovery            [200 usec]
;cnst1: center of aliphatic 1H chemical shifts (offset, in ppm)
;cnst2: 1H-13C J-coupling constant of CH groups (in Hz)
;cnst3: 1H-13C J-coupling constant of CH2 (or CH) groups (in Hz)
;cnst4: power level for 1H-13C het-cp (in Hz)
;l2: loop for DIPSI2 cycle: d8 = ((p13*28.778) * l2)
;inf1: 1/SW(C) = 2 * DW(C)
;in0: 1/(2 * SW(C)) = DW(C)
;nd0: 2
;ns: 2 * n
;ds: 32
;td1: number of experiments in F1
;FnMODE: Echo-Antiecho in F1
;cpdprg1: dipsi2.p13
;cpd2: decoupling according to sequence defined by cpdprg2
;cpdprg4: 2H decoupling sequence applied during t1 and taub [waltz16]
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;pcpd4: f4 channel - 90 degree pulse for 2H decoupling sequence


;use gradient ratio:    gp 1 : gp 2 : gp 3 : gp 4: gp 5: gp 6: gp 7: gp 8
;                         30 :   65 :   20 :  -60:   50:   25    80: 20.1

;for z-only gradients:
;gpz1: 30%
;gpz2: 65%
;gpz3: 20%
;gpz4: -60%
;gpz5: 50%
;gpz6: 25%
;gpz7: 80%
;gpz8: 20.1%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100
;gpnam6: SMSQ10.100
;gpnam7: SMSQ10.100
;gpnam8: SMSQ10.100


;$Id: hcchdigp3d2,v 1.12 2014/02/11 11:31:28 ber Exp $
```